Part 1. 可視化の導入
1 Data import
Excel形式のデータのインポートし、この後の解析に向けて簡単な処理を行ないます。
part1.do
import excel "NHEFS.xlsx", firstrow clear

///// アウトカム変数へのラベル作成 /////
label define death 0 "Alive" 1 "Death"
label values death death

///// pack_year_n変数の作成 /////
gen pack_year_n = (smokeintensity/20) * smokeyrs

compress
save nhefs.dta, replace

また、必要な外部コマンド（user-driven commandまたはcommunity-based command等）をインストールして下さい。既にインストール済みのものは省いても差し支え有りません。
part1.do
ssc install schemepack, replace   // イケてるグラフテーマ
ssc install stripplot, replace    // dotplotの（上位）互換コマンド 
ssc install fre, replace          // 度数分布表。tabの互換コマンド
ssc install table1, replace       // 記述統計表を作成
ssc install bihist, replace       // 上下に向いたヒストグラム
ssc install covbal, replace       // 共変量バランスのチェック
// ssc install dag, replace       // DAGを用いた解析（今回はつかいません）

// グラフを1枚にまとめるときに、Legendを1つだけにするコマンドgrc1leg	
net install grc1leg.pkg, replace from(http://www.stata.com/users/vwiggins/)  

// labmaskのインストール
net install gr0034.pkg, replace from(http://www.stata-journal.com/software/sj8-2/)

// カラーパレット
net install gr0075.pkg, replace from(http://www.stata-journal.com/software/sj18-4/)
3 twoway
3.1 twowayについて
（主に）2つの変数の関係を二次元にプロットするためのコマンドです。Rにおけるggplotのようにレイヤーを追加して、複数のグラフを1つの座標系にプロット可能です。Stataでは、//（スラッシュ2つ）の後にコメントを残す事ができます。また、///（スラッシュ3つ）では、その後にコメントを残しつつ、次の行にコマンドが継続することをしまします。

3.2 基本コマンド
Rにおけるggplot2と同じようにtwowayコマンドだけでは、何も描かれていない真っ白い背景だけが用意されます。ここにどんどんとコマンドでグラフに関する指示を重ねていきます。
part1.do
twoway

3.3 基本コマンド+オプション1
Stataのtwowayコマンドでは、twowayの後に描くグラフを指定します。ここでは散布図を描画するサブコマンドscatterを指定しています。他のグラフはコマンドウィンドウにhelp twowayと入力することで調べることが出来ます。なお、サブコマンドが1つのみの場合は、twowayを省略可能です。
drGraph_Part1.do
twoway scatter sbp dbp

[image: グラフ, 散布図

自動的に生成された説明]

3.4 基本コマンド＋オプション2
描いたプロットについて、Stataでは、コマンドオプション（コンマの後に指定する）で編集します。ここでは3つのオプションを指定しています。

drGraph_Part1.do
twoway scatter dbp sbp, ///
	scheme(white_tableau) /// グラフテーマ
	msymbol(oh)        /// プロットシンボル（ohは小さい中抜きの○の意味）
	mcolor(orange %30)    // プロット色と透明度

[image: グラフ, 散布図

自動的に生成された説明]

3.5 基本コマンド＋オプション3
Stataのコマンドのオプションをさらに付け加え、軸に関する情報を増やしました。

part1.do
twoway scatter dbp sbp, ///
	scheme(white_tableau) /// グラフテーマ
	msymbol(oh)        /// プロットシンボル（ohは小さい中抜きの○の意味）
	mcolor(orange %30)    /// プロット色と透明度
	xtitle("SBP(mmHg)")   /// X軸
	ytitle("DBP(mmHg)")   /// Y軸
	xlabel(80(40)240)        /// X軸の範囲
	ylabel(40(20)140)        //  Y軸の範囲

[image: グラフ, 散布図

3 tidyなデータ作成について
StataではRのtidyverseに相当するようなパッケージ群は組まれていないようです。



Part 2: 記述統計の可視化
2 Data
データの概要を確認するためのコマンドです（いずれもStata1.0から存在する古参コマンドです）。
part2.do
des
sum
list in 1/10

3 Overview & Missing values（欠測値）
3.1 Overview
Stataで欠損値の概要を示すためにはmisstableコマンドを利用します。コマンドの詳細はhelp misstableにて確認できます。Rでの実行結果とは異なり、欠損値の状況は表として示されます。
part2.do
misstable sum
misstable pattern
misstable tree
misstable nested

3.2 Missingness at each variable
Stataでは標準でlollipop plotを行なうためのコマンドが存在しませんし、調べた限りでは外部コマンドでもlollipop plotをサポートしているコマンドがありませんでした。そこで、欠損値割合をlollipop plotとして表示するプログラムを作成しました（missing_lollipop_plot.do）。
近日中に、外部コマンド（ado file）として作成公開しようと考えています。中身は100行くらいになるので、ここでは全体を記載していません。
part2.do
qui: do missing_lollipop_plot // program定義の読込み
missing_lollipop_plot

[image: コンピューターのスクリーンショット

3.3 Missingness relationship between variables
3.3.1 全ての変数
Rにおけるvis_miss関数に相当するStata標準コマンドはありませんでしたが、外部コマンドとしてmissingplotがありました。これはNicolas J Cox先生が作成されたものですが、若干古いので、プロットのビジュアルがvis_miss関数よりも見栄えしないことが欠点です。
missingplotコマンドは標準では、欠損値がある変数だけをプロットします。

part2.do
missingplot, scheme(white_tableau) msymbol(oh)

[image: ]
missingplotコマンドでは、残念ながらプロットサイズを小さくしたとしても、欠損値が多い変数（yrdth）などでは、全欠損しているように見えてしまいます。
次に、死亡の有無（death）で層別化した場合に、特定の変数（alcoholhowmuch）の欠損の割合に違いがあるかを確認します。

3.3.2 連続変数 × 名義変数
Stataでは、misstable  sumコマンドを用いて、欠損値数を表示させることができます。ここでifを繋げる事で、死亡の有無で層別します。
part2.do
misstable sum alcoholhowmuch if death==0
misstable sum alcoholhowmuch if death==1

生存では欠損値が308件、死亡では欠損値が109件であることが示されます。misstableコマンドでは割合は計算されませんので、手計算で行なう必要があります。

3.3.3 連続変数 × 連続変数
次は、2つの連続変数間について、一方が欠損値の場合にもう一方の変数の分布がどのようになっているかを確認します。

part2.do 
set seed 12345

gen cholesterol2 = cholesterol
recode cholesterol2 (.=0)
capture graph drop *

// 生存者におけるグラフ
twoway ///
	 (scatter cholesterol2 pack_year_n ///
		if death ==0 & cholesterol!=., jitter(2 2)) ///
	(scatter cholesterol2 pack_year_n ///
		if death ==0 & cholesterol==., jitter(2 2) ///
	ylabel(0(100)400) xlabel(0(50)150) ///
	scheme(white_tableau) legend(off) title("Alive") name(foobar1, replace))

// 死亡者におけるグラフ
twoway ///
	 (scatter cholesterol2 pack_year_n ///
		if death ==1 & cholesterol!=., jitter(2 2)) ///
	(scatter cholesterol2 pack_year_n ///
		if death ==1 & cholesterol==., jitter(2 2) ///
	ylabel(0(100)400) xlabel(0(50)150) ///
	scheme(white_tableau) legend(off) title("Death") name(foobar2, replace))

// 2つのグラフを結合
graph combine foobar1 foobar2, /// 
cols(2) name(chol_smk, replace) scheme(white_tableau)

scatterコマンドで散布図のプロットを行なっています。欠損値の有無、死亡の有無で分けて描画していますので、都合4回scatterを行い、それらを結合させています。jitterオプションで撒布をランダムに散らしていますので、set seed 12345によって乱数固定をかけなければ、毎回微妙に異なるグラフになります。
[image: グラフ, 散布図

4 Continuous variable (連続変数)
4.1 ヒストグラム
Stataではhistogramサブコマンドでヒストグラムを描画します。
part2.do
twoway histogram pack_year_n, ///
	start(0) width(5) freq /// BINの開始、BIN幅、縦軸の指定
	scheme(white_tableau) /// グラフテーマ
	xtitle("Pack-Year") /// 横軸タイトル
	name(hist, replace)

[image: グラフ, ヒストグラム

自動的に生成された説明]
4.2 密度関数プロット
Stataではkdensityサブコマンドで密度関数プロットを描画します。kdensityサブコマンドを2回使用しています。一回目ではif death==0を指定し、二回目でif death==1を指定し、生存者・死亡者のグラフを描いています。
part2.do
twoway ///
(kdensity pack_year_n if death==0) /// 生存者のグラフ
(kdensity pack_year_n if death==1, /// 死亡者のグラフ
scheme(white_tableau) /// グラフテーマ
legend(label(1 "Alive") label(2 "Death")) /// 凡例の指定
xtitle("Pack-Year") /// X軸タイトル
ytitle("Density") /// Y軸タイトル
name(dens, replace))

[image: グラフ, 折れ線グラフ

4.3 箱ひげ図
Stataでの箱ひげ図は、ここまでで利用してきたtwoway系列のコマンドとは別に用意されています。graphコマンドのサブコマンドboxを利用して描画します。

graph box pack_year_n , /// 
	scheme(white_tableau) /// グラフテーマ
	ytitle("Pack-Year") /// Y軸
	over(death) /// deathによってグラフを分ける。なおby(death)も可
	name(box_over, replace)
[image: グラフ, 箱ひげ図

箱ひげ図にドットプロットを載せるには、通常のコマンドでは難しく、外部コマンドstripplotを使用する必要があります。このコマンドを用いるとドットプロットに箱ひげ図を載せるイメージでグラフが作成できます。

stripplot pack_year_n, ///
	scheme(white_tableau) /// グラフテーマ
	vertical /// 縦向きのグラフ
	ytitle("Pack-Year") /// Y軸
	over(death) /// deathによってグラフを分ける
	box /// BOXプロットを追加
	jitter(30 10) /// ドットを散らす
	name(box_dot, replace)


[image: ダイアグラム, 設計図

4.4 ドットプロット
Stataにおいてドットプロットの描画には、いくかの方法があります。1つめは上記で使用したstripplotコマンドを利用する方法です。その他にdotplotコマンドやgr7コマンドがあります。dotplotコマンドはStata社公式のコマンドですが、やや柔軟性に欠けています。gr7は公式の古いコマンドですので、利用している人があまりいませんし、やや時代がかったグラフが描画されます。
この中ではstripplotコマンドが一番柔軟に描画可能です。例えば、縦向き（vertical）グラフと横向き（horizontal）グラフの両方が描画できるのはstripplotコマンドのみです。
ここでは、stripplotコマンドを用いた描画方法を紹介します。

4.4.1 何も施していないドットプロット
stripplot pack_year_n, sheme(white_tableau) over(death) vertical
[image: テーブル

4.4.2 ポイントサイズを小さくしたドットプロット
ポイントサイズの調整は、msizeオプションで実行します。
part2.do
stripplot pack_year_n, sheme(white_tableau) over(death) vertical msize(0.5) 
[image: テーブル

4.4.3 ポイントを少しずらしたジッタープロット
ジッタープロットを行なうためには、jitterオプションを用います。
part2.do
stripplot pack_year_n, ///
sheme(white_tableau) over(death) vertical jitter(5 5) center
[image: テーブル

4.4.4 ビースウォーム（っぽい）プロット
Rで実行されていたように、確率分布に基づいてずらしたBeeswarmプロットはサポートされていません。が、オプションの組み合わせにより、それっぽいプロットを行なうことが出来ます。ここではstackオプションを用いることにより、ビースウォームプロットに似せました。

part2.do
stripplot pack_year_n, ///
scheme(white_tableau) over(death) vertical center ///
stack

[image: グラフ

4.5 レインクラウドプロット
Rain cloudプロットについてStataで実行可能かどうか？について、2022年11月20日に、Statalistで質問上がっていました。Nicolas J Cox先生の回答では「It's a hybrid box plot, density trace and parallel coordinates (profile, slope) plot. You need twoway for that if it has not been written up as a single command before.」とのことであり、単一コマンドでの描画はサポートされていないようです。Lollipop plotと異なって、私も作成方法がすぐに思いつきませんでした。
なお、回答者のNicolas J Cox先生はここまでで使用した外部コマンドmissingplotやstripplotの作成者です。
https://www.statalist.org/forums/forum/general-stata-discussion/general/1690233-raincloud-plot-in-stata

5 Nominal variable（名義変数）
5.1 棒グラフ
5.1.1 ラベルなし
Stataでは、graphコマンドのサブコマンドbarを用いて棒グラフを描画します。ここでは変数seqnについて数え上げ(count)をするように指定しています。
part2.do
graph bar (count) seqn, scheme(white_tableau) over(marital) ytitle("count")

[image: グラフ, 棒グラフ

5.1.2 ラベルあり
ラベルありの棒グラフにするためには、変数に値ラベルをつける必要があります。
part2.do
label define marital /// 値ラベルの定義
1 "Under17" 2 "Married" 3 "Widowed" ///
4 "Never" 5 "Divorced" 6 "Separated" 8 "Unknown"
label values marital marital /// 値ラベルを変数に割り当て
graph bar (count) seqn, scheme(white_tableau) over(marital) ytitle("count")
[image: グラフ, 棒グラフ

5.2 横並び棒グラフ
グラフに層別にはoverオプションを利用します。このoverオプションは複数回利用することが可能です。このオプションの出現順をかえると、また異なるグラフになります。
part2.do
graph bar (count) seqn, ///
scheme(white_tableau) ///
over(death) over(marital) /// overオプションを重ねて利用
ytitle("count")

[image: グラフ, 棒グラフ

5.3 100％積み上げ棒グラフ
今回は、横向きの棒グラフですので、graphコマンドのサブコマンドhbarを用います。また、積み上げのためのオプションとしてstackを用いています。積み上げる中身（死亡数と生存数）を予め算出した後にグラフ描画に入っていますので、やや手間がかかっています。
part2.do
label define education /// 値ラベルの定義
5 "College or more" 4 "College dropout" 3 "HS" ///
2 "HS dropout" 1 "8th grade or less"
label values education education // 値ラベルの割当て

bys education: egen frac_1 = sum(death) // 各教育における死亡数
bys education: gen frac_3 = _N // 各教育における人数
gen frac   = frac_1/frac_3 // 各教育における死亡割合
gen frac_2 = frac_3 - frac_1 // 各教育の生存数

graph hbar (mean) frac_2 frac_1, /// 生存数と死亡数で積み上げる
scheme(white_tableau) ///
over(education, sort(frac)) /// 各教育のバーをつくり、死亡割合でソートする
stack /// 積み上げグラフにする
percent /// 100％の積み上げにする。
legend(label(1 "Alive") label(2 "Death"))

[image: グラフ, 棒グラフ

6 Multivariable（多変量）
Stataにおいて、利用する変数同士の関係を網羅的に可視化するためのプロットとしては、graph matrixがあります。ただし、これは散布図を描くのみですので、Rのggpairs関数とは異なって変数型に合わせてグラフを調整するという機能はありません。下記では一部変数でのみを用いて作図しています。

graph matrix sbp dbp age income, ///
scheme(white_tableau) ///
half /// 散布図を片側だけにした
msymbol(smcircle_hollow)

[image: テーブル

Part 3 推測統計の可視化
2 Data source
別途、Completeデータをcsvファイルから読み込み、必要な変数のみにまとめたり、値ラベルを定義したりします（このCompleteデータのcsvファイルは、Rから呼び出したファイルです）。
part3.do
* Data source
import delimited "nhefs_complete.csv", clear

// 必要変数のみ残す
keep seqn death yrdth modth dadth sex age race education exercise ///
	smokeintensity smokeyrs

// label
label define sex 0 "Male" 1 "Female"
label values sex sex

label define race 0 "White" 1 "Back or other"
label values race race

label define edu 1 "8th grade or less" 2 "HS dropout" 3 "HS" 4 "College dropout" 5 "College or more"
label values education edu

label define exe 0 "Much exercise" 1 "Moderate exercise" 2 "Little or no exercise"
label values exercise exe
4 Data
4.1 Outcome
Rから出力したcsvをStataで読み込むと、欠損値（NA）が文字列として扱われるため、注意が必要です。ここでは、それを逆手に取って、文字列型のまま死亡日（tmp1）を作成しています。そのtmp1をdate関数でSIF形式に変換しています。
part3.do
* 死亡日・打切り日作成
gen tmp1 = "19" + yrdth + "/" + modth + "/" + dadth if death==1
gen tmp2 = date(tmp, "YMD")
gen event_date = cond(death==0, td(31Dec1992), tmp2)
drop tmp*
format %td event_date

* 観察開始日からの日数を計算
gen survtime = datediff(td(01Jan1983), event_date, "day")

4.2 Exposure
genコマンドで変数を作成し、labelコマンドで値ラベルを定義し、貼り付けています。
part3.do
gen pack_years_n = (smokeintensity / 20) * smokeyrs
gen pack_years = cond(pack_years >= 20, 1, 0)
label define pack_years 0 "Low" 1 "High"
label values pack_years pack_years

4.3 練習のためのデータ変更
Rでの操作をトレースしていますが、乱数の出目が異なるため、同一データにはなっていません。
part3.do
* 4.3 練習のためのデータ変更
set seed 1234

gen     death1     = rbinomial(1, 0.6)
gen     death0     = rbinomial(1, 0.4)
gen     death2     = cond(pack_years == 1, death1, death0)
gen     censor     = rbinomial(1, 0.2)
gen     r_time     = ///
cond((censor == 1 & death == 0) | (death2 == 1 & death == 0), ///
round(runiform(0, 3652)),0)
replace death      = cond(death2 == 1, 1, death)
replace survtime   = survtime - r_time
gen     survtime_y = survtime/365.25

4.4 本パートで仮定した因果構造
RではDAGを描画していますが、Stataでは（調べた限り）DAGを描画するためのコマンドがありませんでした。DAGをベースにした解析を行なう外部コマンドdagを作成したチームが下記のコメントをしています。
We are working on A Stata command drawing a high-quality DAG.
https://medical-statistics.dk/MSDS/epi/dag/dag.html

5 統計解析
5.1 背景情報の要約
外部コマンドtable1を用いて背景情報をまとめることができます。table1はvarsオプション内で記述する変数とその型をしていしています。byオプションで指定した変数で分けて記述統計量を算出します。
part3.do
table1, vars(sex bin \ age conts \ race bin \ education cat \ exercise cat) by(pack_years) test

5.2 アウトカムの分布の確認
Kaplan-Meier曲線を描いています。Stataではsts graphで生存時間分析におけるグラフ描画が行えます。リスク表を作成するためにrisktableオプション、MSTを図示するためのaddplotオプションのために、かなり長大になっています。
part3.do
// Survival analysis実行用にデータ設定
stset survtime_y, fail(death==1)

// HighにおけるMSTを先に算出してグラフで利用
qui stsum if pack_years==1
local high_p50 = `r(p50)'

// グラフ描画
sts graph, ci scheme(white_tableau) by(pack_years)  ///
	risktable(0(2)10, rowtitle(Low)  failevents ///
		size(small) group(#1) title("At risk (Events)", size(small))) ///
	risktable(0(2)10, rowtitle(High) failevents size(small) group(#2)) ///
	xtitle("Years") ytitle("Overall survival probability") ///
	title("Kaplan-Meier Curve", size(small)) /// 表題など
	censored(single) /// 打切りの印を付ける
	legend(position(12)) /// legendを12時方向に
	addplot(function y=`high_p50', ///
		horizontal range(0 0.5) lcolor(gs8) lpattern(shortdash) || /// 
		function y=0.5, ///
		range(0 `high_p50') lcolor(gs8) lpattern(shortdash) legend(row(1) ///
	order(2 "Low" 1 "95%CI" 4 "High" 3 "95%CI"))) ///
	name(KMC_unwt, replace)

[image: グラフ, 折れ線グラフ

リスクテーブルでは、カッコ内の数字が、その区間のEvent発症者数になります。また、打切り者数をリスクテーブルに表示する方法は、（調べた限り）ありませんでした。グラフ中に打切り者数を表示させることは可能ですが、数字が見にくくなってしまうため、採用していません。

5.3 喫煙の程度と死亡との関連評価
5.3.1 生存時間中央値
下記はStataでは、stset後でのみ有効なコマンドです。今回はグラフ描画の最初でstsetを実行していますので、有効です。オプションを変更する事で、中央値以外のNパーセンタイルを算出することもできます。
part3.do
stsum, by(pack_years) // 点推定値など
stci , by(pack_years) // MSTとその信頼区間

5.3.2 群の時点ごとの生存確率と95%信頼区間
part3.do
sts list, risktable(0/9) by(pack_years) enter

Rでの実行結果とは異なり、時点ゼロでのAt risk人数がゼロ人になっています。代わりに、enterオプションを指定しているのでEnterという列があり、時点ゼロで744人（822人）が追跡コホートに入ったことを示しています。

5.3.3 生存確率の群間比較
Log-rank検定を実行します。
part3.do
sts test pack_years

5.4 喫煙の程度の死亡への影響評価
5.4.2 逆確率重み付けを用いた群間バランスの調整
Stataでは、個人ごとの曝露確率（pack_years=1になる確率）をlogistic modelで算出します。通常であれば、xiプレフィクスコマンドは使わなくても問題ありません。このxiプレフィクスコマンドを付けることによって、一時的に作成されたダミー変数を自動削除せずに残しておくことができます（今回は後で使うので残しています）。
part3.do
xi:logit pack_years i.sex age i.race i.education i.exercise
predict ps, pr

gen iptw = cond(pack_years==1, 1/ps, 1/(1-ps))

次に傾向スコアの分布を確認する貯めに、ヒストグラムを描画しますが、Stataの標準では、Rで作成されたような上下に分かれるヒストグラムは作成できません。今回は外部コマンドbihistを利用して、作図します。
まず、重みを付けていない場合のヒストグラムを描画します。bihistコマンドはやや特殊で、描画の外見に関するオプションをtwオプションの中に入れて記載する必要があります。
part3.do
bihist ps, by(pack_years) percent width(0.05) start(0) ///
	tw(scheme(white_tableau)  xtitle(Propensity Score) ytitle(Percent) ///
		legend(order(2 1))) ///
	name(unwt_hist, replace)

[image: グラフ, 棒グラフ

次に、重みを付けてたグラフを描画します。重みを指定するための記述（[pw=iptw]）を追記します。
part3.do
bihist ps [pw=iptw], by(pack_years) percent width(0.05) start(0) ///
	tw(xtitle(Propensity Score) ytitle(Percent) ///
		legend(order(2 1))) ///
	name(iptw_hist, replace)

[image: グラフ, 棒グラフ

これら2つのグラフを1つにまとめることで、比較しやすいグラフを作成しようと思います。標準コマンドであれば、graph combineで結合するところですが、Legendが重複してしまうという欠点があります。一般には、結合する片方でLegendを付けないという対策で回避しますが、ここでは、外部コマンドgrc1legを利用します（Graph Combine 1 Legendの略）。
オプションycommonはy軸を2つのグラフで同じスケールにするという指定で、positon(3)はLegendを3時方向に入れるという指定です。
part3.do
grc1leg unwt_hist iptw_hist,  ycommon position(3) name(comb_hist, replace)
[image: グラフ, 棒グラフ

重みの大きさの確認を行います。これは、bysortプレフィックスを付け、summarizeコマンドで重み変数iptwを表示させていまます。なお、bysortはbys、summarizeはsuで略しています。
part3.do
bys pack_years: su iptw, detail

次に、交絡要因ごとに群間のバランスを確認します。ここでは外部コマンドのcovbalを用います。なお、この結果がRで行われた結果が一部異なっています。この原因・理由については検証できていません。
part3.do
covbal pack_years ps _Isex_1 age _Irace_1 _Ieducation_2 _Ieducation_3 ///
_Ieducation_4 _Ieducation_5 _Iexercise_1 _Iexercise_2, ///
saving(bal_unwt, replace)

covbal pack_years ps _Isex_1 age _Irace_1 _Ieducation_2 _Ieducation_3 ///
_Ieducation_4 _Ieducation_5 _Iexercise_1 _Iexercise_2, ///
wt(iptw) saving(bal_iptw, replace)
ここで、傾向スコア算出の際に、xiをプレフィックスにつけたことが生きてきます。_Iから始まる変数は、xiによって生成された変数です。_Ieducation_1は変数education=1のダミー変数（つまり、8th grade or less）を示しています。他も同様です。
3カテゴリ以上ありカテゴリ変数において標準化差を算出するには、2つ方法があり、1つはダミー変数化したものを二値変数としてそれぞれについて標準化差を計算うする方法で、もう1つは専用の公式を用いる方法です。Ｒのlove.plot関数では1つめの選択をとっていますので、ここでもそれに合わせました。
1つめのcovbalコマンドでは、重み付け前のバランスを標準化差と分散比で表示させます。2つめのcovbalでは、3行目にwt(iptw)というオプションが入っていますが、これが変数iptwで重み付けた後のバランス評価を行うという指示になります。いずれのcovbalにもsavingオプションを指定しています。これは、次に可視化するときに役に立つので、得られた結果表をデータとしてセーブするためのものです。

変数の群間バランスを図示するための標準コマンドおよび外部コマンドは調べて限りありませんでした。しかし、covbalコマンドで出力したデータを元にすれば、Stataでも散布図の応用で作図可能です。

part3.do
capture frame drop covbal
frame create covbal
frame covbal {
	* データセット加工
	use bal_unwt, clear
	gen odr = _N - _n + 1
	keep odr varname stdiff varratio
	rename stdiff stdiff_unwt
	rename varratio varratio_unwt
	
	merge 1:1 varname using bal_iptw
	assert _merge == 3 // _merge==3以外があれば、何かおかしいのでassertで止める
	drop _merge tr_mean tr_var tr_skew con_mean con_var con_skew
	rename stdiff stdiff_iptw
	rename varratio varratio_iptw
	sort odr
	
	labmask odr, value(varname)
	
	* カラーパレット設定
	colorpalette hcl, select(1 6 9) nograph
	local unadj `r(p1)'
	local adj `r(p3)'
	local zero `r(p2)'
	
	* 標準化差のグラフ
	twoway ///
	catter odr stdiff_unwt, ylabel(10(1)1, valuelabel) mcolor("`unadj'") || ///
	scatter odr stdiff_iptw, ylabel(10(1)1, valuelabel) mcolor("`adj'") || ///
	function y= 0.1, horizontal range(0 10) lcolor(gs8) lpattern(shortdash) || ///
	function y=-0.1, horizontal range(0 10) lcolor(gs8) lpattern(shortdash) || ///
	function y=0   , horizontal range(0 10) lcolor("`zero'")  ///
	legend(order(1 "Unadjusted" 2 "Adjusted")) ///
	xtitle("Standardized Mean Difrences") title(Covariate Balance) ///
	name(bal_smd, replace)
	
	* 分散比のグラフ
	twoway ///
	scatter odr varratio_unwt, ylabel(10(1)1, valuelabel) mcolor("`unadj'") || ///
	scatter odr varratio_iptw, ylabel(10(1)1, valuelabel) mcolor("`adj'")|| ///
	function y=1.25, horizontal range(0 10) lcolor(gs8) lpattern(shortdash) || ///
	function y= 0.8, horizontal range(0 10) lcolor(gs8) lpattern(shortdash) || ///
	function y=   1, horizontal range(0 10) lcolor("`zero'")  ///
	legend(order(1 "Unadjusted" 2 "Adjusted")) ///
	xscale(log) xlabel(0.5 0.6(0.2)1.0 1.25) xscale(extend) ///
	xtitle("Variance Ratio, log-scale") title(Covariate Balance) ///
	name(bal_vr, replace)
}

作図はtwowayで5つのレイヤーを重ねています。1つめが重み付け無しの場合のプロット、2つめが重み付けた時のプロット、3つめが閾値の上限の縦ライン、4つめが閾値の下限の縦ライン、5つめが中央の縦ラインです。閾値ラインの設定はfunction y=0.1などの数値を変えることで変更できます。
標準化差のプロットは下記の様になります。
[image: テーブル

分散比のプロットは下記の様になります。横軸を対数軸にしていることに注意して下さい。
[image: グラフ, 折れ線グラフ, 散布図

5.4.3 推定
ATEを推定します。Stataで生存時間分析を実施する際には、生存時間分析のための設定を先に指定しておく必要があります。そのための設定がstsetコマンドです。時間を表す変数、重み、イベントを示す変数を設定しました。次にstcoxコマンドでCox比例ハザードモデルによる解析が行うことができます。
part3.do
// Survival analysis実行用にデータ設定（iptwで重みを付ける）
stset survtime_y [pw=iptw], fail(death==1)

// Cox比例ハザードモデル
stcox pack_years

乱数の出目がRとはことなったり、原因不明ですがIPTWによる共変量バランスの調整具合が異なったりしていますので、ハザード比についても、Rでの結果（点推定値1.46）とは異なり1.40（95%CI: 1.17, 1.68）となりました。
この重み付けによって交絡を調整した生存確率について、可視化します。基本的に重み付けを行う前と同じ操作を行っていますが、ciオプションを削除しています。これは、Stataによる生存確率のグラフでは、stsetで重み付けを指定していると信頼区間の描画がでないためです。また、イベント数を表示することは出来なくはありませんが、重み付けであるため、イベント数には小数点下の数値もあります。いろいろと工夫はしたのですが、小数点下の桁数の表記が上手くいかず、諦めました。

part3.do
// HighにおけるMSTを先に算出してグラフで利用
qui stsum if pack_years==1
local high_p50 = `r(p50)'

// グラフ描画本体
sts graph, scheme(white_tableau) by(pack_years)  ///
	risktable(0(2)10, format(%9.1f) rowtitle(Low) ///
		failevents format(%9.1f) size(small) group(#1) ///
		title("At risk", size(small))) ///
	risktable(0(2)10, format(%9.1f) rowtitle(High) ///
		failevents size(small) group(#2)) ///
	xtitle("Years") ytitle("Overall survival probability") ///
	title("Kaplan-Meier Curve", size(small)) ///
	censored(single) ///
	legend(position(12)) ///
	addplot(function y=`high_p50', horizontal range(0 0.5) ///
		lcolor(gs8) lpattern(shortdash) || ///
		function y=0.5, range(0 `high_p50') lcolor(gs8) ///
		lpattern(shortdash) legend(row(1) ///
	order(1 "Low" 2 "High"))) name(KMC_iptw, replace)
[image: グラフ, 折れ線グラフ

